Herramientas para el diseño y análisis de redes de transporte urbano de pasajeros

Tema 5: Modelo básico de optimización de redes de transporte urbano (cont.)

Modelo básico

- Variantes que optimizan solo costos variables y solo costos fijos, con y sin restricciones.

- Ejemplos numéricos con un grafo concreto y dos mercancías: (1) 10 unidades del nodo 1 al 6 y (2) 5 unidades del nodo 4 al 6.



Optimizar solo costos variables, sin restricción de presupuesto

$$
\begin{aligned}
\min \quad & \sum_{k \in K} \sum_{a \in A} c_a x_{ak} & \\
\text{s.a.} \quad & \sum_{a \in A_n^+} x_{ak} - \sum_{a \in A_n^-} x_{ak} = \theta_{nk} & \forall\ n \in N, k \in K, \\
& x_{ak} \leq R_k y_a & \forall\ a \in A, k \in K, \\
& x_{ak} \geq 0 & \forall\ a \in A, k \in K, \\
& y_a \in \{0, 1\} & \forall\ a \in A.
\end{aligned}
$$

Resultado

- Se habilitan los arcos 1_3, 1_2, 3_5, 2_4, 5_6, 4_6, variables y.
- Pero solamente hay flujo sobre los arcos 1_3, 3_5, 5_6 (10 unidades de mercancía 1) y 4_6 (5 unidades de mercancía 2), variables x.



Optimizar solo costos variables, con restricción de presupuesto

$$
\begin{aligned}
\min \quad & \sum_{k \in K} \sum_{a \in A} c_a x_{ak} & \\
\text{s.a.} \quad & \sum_{a \in A_n^+} x_{ak} - \sum_{a \in A_n^-} x_{ak} = \theta_{nk} & \forall\, n \in N, k \in K, \\
& \sum_{a \in A} f_a y_a \leq B, & \\
& x_{ak} \leq R_k y_a & \forall\, a \in A, k \in K, \\
& x_{ak} \geq 0 & \forall\, a \in A, k \in K, \\
& y_a \in \{0, 1\} & \forall\, a \in A.
\end{aligned}
$$

Resultado

- Para presupuesto $B = 16$, se habilitan los arcos 1_3, 3_5, 5_6, 4_6.



Resultado

- Para presupuesto $B \in [12, 16)$, se habilitan los arcos 1_3, 3_4, 4_6. La mercancía 1 debe viajar por un camino más largo respecto a su ideal (costo 12 vs. costo 9).

- Para presupuesto $B < 12$ el problema no es factible.



Optimizar solo costos fijos, sin restricción de costo variable

$$\begin{aligned}
\min \quad & \sum_{a \in A} f_a y_a \\
\text{s.a.} \quad & \sum_{a \in A_n^+} x_{ak} - \sum_{a \in A_n^-} x_{ak} = \theta_{nk} && \forall\ n \in N, k \in K, \\
& x_{ak} \leq R_k y_a && \forall\ a \in A, k \in K, \\
& x_{ak} \geq 0 && \forall\ a \in A, k \in K, \\
& y_a \in \{0, 1\} && \forall\ a \in A.
\end{aligned}$$

Resultado

- Se habilita el subconjunto menos costoso de arcos que mantienen conectados los orígenes y los destinos de cada una de las mercancías.



Optimizar solo costos fijos, con restricción de costo variable

$$
\begin{aligned}
\min \quad & \sum_{a \in A} f_a y_a && \\
\text{s.a.} \quad & \sum_{a \in A_n^+} x_{ak} - \sum_{a \in A_n^-} x_{ak} = \theta_{nk} && \forall\ n \in N, k \in K, \\
& \sum_{a \in A} c_a x_{ak} \leq \rho c_k^* R_k && \forall\ k \in K, \\
& x_{ak} \leq R_k y_a && \forall\ a \in A, k \in K, \\
& x_{ak} \geq 0 && \forall\ a \in A, k \in K, \\
& y_a \in \{0, 1\} && \forall\ a \in A.
\end{aligned}
$$

Resultado

- En este caso, $c_1^* = 9$ y $c_2^* = 7$.

- Para $\rho \geq 12/9$ se obtiene la misma solución que en el caso anterior, dado que el valor del parámetro permite que las mercancías se desvíen de su camino ideal.



Resultado

- Para $\rho < 12/9$ se obtiene la solución donde las mercancías transitan por su camino ideal.

- Si bien la función objetivo busca minimizar los costos fijos, la restricción de costos variables impone un mínimo nivel de servicio.



Modelo básico en GLPK

```
set N; # nodos
set A; # arcos
set K; # mercancias

set salientes{N} within A; # arcos salientes de los nodos de N
set entrantes{N} within A; # arcos entrantes a los nodos de N

param c{a in A} >= 0; # costo variable del arco a
param f{a in A} >= 0; # costo fijo del arco a

param O{k in K}; # nodo origen de mercancia k
param D{k in K}; # nodo destino de mercancia k
param R{k in K}; # cantidad de demanda de mercancia k
param S{k in K}; # costo del camino mas corto de mercancia k
```

Modelo básico en GLPK (cont.)

```
param B; # presupuesto
param rho; # desvio permitido del camino mas corto para cualquier mercancia

var x {a in A, k in K} >= 0; # flujo de mercancia k en el arco a
var y {a in A} binary; # habilitacion de arco a

minimize costo: sum {a in A, k in K} (c[a] * x[a,k]) +
                sum {a in A} (f[a] * y[a]);
# funcion objetivo generica

s.t. conservacion_flujo {n in N, k in K}:
sum {a in salientes[n]} x[a,k] - sum {a in entrantes[n]} x[a,k] =
if n=O[k] then R[k] else (if n=D[k] then -R[k] else 0);

s.t. activacion {a in A, k in K}: x[a,k] <= R[k] * y[a];
```

Modelo básico en GLPK (cont.)

Restricciones particulares de cada variante.

```
s.t. presupuesto: sum {a in A} f[a] * y[a] <= B;
# restriccion de presupuesto

s.t. costo_var {k in K}: sum {a in A} (c[a] * x[a,k]) <=
                         rho * S[k] * R[k];
# restriccion de costo variable por mercancia
```

Datos de instancia

```
data;

set N := 1 2 3 4 5 6;

set A := 1_3 1_2 3_5 3_4 2_4 2_3 5_4 5_6 4_6;

set K := 1 2;

set salientes[1] := 1_2 1_3;
set salientes[2] := 2_3 2_4;
set salientes[3] := 3_4 3_5;
set salientes[4] := 4_6;
set salientes[5] := 5_4 5_6;
set salientes[6] := ;

set entrantes[1] := ;
```

```
set entrantes[2] := 1_2;
set entrantes[3] := 1_3 2_3;
set entrantes[4] := 2_4 3_4 5_4;
set entrantes[5] := 3_5;
set entrantes[6] := 4_6 5_6;

param:    c :=
1_3 4
1_2 6
3_5 2
3_4 1
2_4 2
2_3 2
5_4 1
5_6 3
4_6 7
;
```

```
param:    f :=
1_3 4
1_2 6
3_5 2
3_4 1
2_4 2
2_3 2
5_4 1
5_6 3
4_6 7
;

param:    O :=
1 1
2 4
;

param:    D :=
```

```
1 6
2 6
;

param:    R :=
1 10
2 5
;

param:    S :=
1 9
2 7
;

param B := 16;

param rho := 1;
```

```
end;
```